Program report

MACZ
Universidad Nacional Autónoma de México

27 de diciembre de 2014

Túnel circular con recubrimiento. Ambos materiales elásticos e isótropos. Tomado del capítulo 3 de Mow y Pao, "The diffraction of elastic waves and dynamic stress concentrations", 1971. Ejercicio para incidencia de onda plana P y SV.



Fig. 4.4. Elastic Cylinder

```
module datos
save
real*8, dimension(2), parameter :: rho = (/1.45, 1.0/)
real*8, dimension(2), parameter :: nu = (/0.25, 0.20/)
real*8, dimension(2), parameter :: bet = (/2.405, 1.700/) * 10
real*8, dimension(2) :: alf
integer,parameter :: nRes = 2
real*8, dimension(2+nRes) :: radios ! a, b
real, parameter :: Qq = 1000.0, periodicdamper = 1.0, TW = 1.0
real*8, parameter :: DFREC = 0.05
integer, parameter :: NFREC = 100, nplot = 100
integer, parameter :: nmax = 500, nfracs = 1
real*8, parameter :: Dns = 1.
complex*16, parameter :: UI = cmplx(0.0d0,1.0d0,8), &
                         UR = cmplx(1.0d0,0.0d0,8), &
                         Z0 = cmplx(0.0d0,0.0d0,8)
real*8, parameter :: PI = real(4.0d0*ATAN(1.0d0),8)
contains
subroutine set_radios
radios(1) = 3.00 ! a
```

```
      radios(2) = radios(1)*1.1 ! b
      end subroutine set_radios
      end module datos
```

```
        Rw(1)%r = radios(2) !r=b in the medium at the boundary of the liner
        Rw(1)%th = pi!/2.
        Rw(1)%reg = 1 ! in the medium

        Rw(2)%r = radios(1) !r=a at the inner side of the cylinder
        Rw(2)%th = pi!/2.
        Rw(2)%reg = 2 ! in the liner
```

```
      PROGRAM tunelRecub
! Solución analítica de la difración de una onda P incidente
! en una cavidad cilíndrica circular con recubrimiento
      use datos; use vars_func_of_w; use Hank
      use debug; use RES; use plotter
      implicit none
      integer :: J,n,et,info,i
      real*8, pointer :: r,th
      integer, pointer :: reg
      complex*16, dimension(6,6) :: M
      complex*16, dimension(6,2) :: B !P , SV
      integer, dimension(6) :: ipiv
      complex*16,dimension(nplot) :: pt_RES
      ! terms defined in the appendix (functions)
      complex*16 :: e11,e12,e21,e22,e41,e42,e71,e72,e81,e82,sig0
      real*8,dimension(2) :: BEALF
      call set_radios !a,b
      call setup_resu !puntos receptores
!     OMEI = - periodicdamper*PI/TW
      eta = radios(2)/radios(1) !b/a
      BEALF(1:2)=SQRT((0.5-NU(1:2))/(1.0-NU(1:2))) !IF POISSON RATIO IS GIVEN
      alf(1:2) = bet(1:2)/BEALF(1:2)
```

```
      do J=1,NFREC !*********************************************
      FREC=DFREC*real(J); if (J .eq. 1)  FREC = 0.5_8 * DFREC ! Hz
      OME=2.0*PI*FREC !rad/s
!     COME = CMPLX(OME, OMEI,8)
!     VEL(1,1:2) = cmplx(alf(1:2)*&
!                  (1.+1./pi/Qq*log(ome/2./pi)),-1./2./Qq,8)
!     VEL(2,1:2) = cmplx(bet(1:2)*&
!                  (1.+1./pi/Qq*log(ome/2./pi)),-1./2./Qq,8)
```

```
COME = OME * CMPLX(1., -1./(2.*Qq),8)
VEL(1,1:2) = cmplx(alf(1:2),(/0.,0./),8)
VEL(2,1:2) = cmplx(bet(1:2),(/0.,0./),8)
cp(1:2) => VEL(1,1:2) ! dilatación
cs(1:2) => VEL(2,1:2) ! corte
w_c(1:2,1:2) = cOME / VEL(1:2,1:2)
beta(1:2) => w_c(2,1:2) !shear wave number
alfa(1:2) => w_c(1,1:2) !compressional wave number
p2(1:2) = (alfa(1:2))**2.0 !compressional wave number (square)
s2(1:2) = (beta(1:2))**2.0 !shear wave number (square)
aMU(1:2) = RHO(1:2) * cs(1:2)**2.
Lambda(1:2) = RHO(1:2)* cp(1:2)**2. - real(2.)* aMU(1:2)
muR = amu(1)/amu(2) !shear moduli ratio
gammaP = z0! spacing variable  2.5D
gammaS = z0! spacing variable
call makeBessels(.false.) ! n = -1,0,1,2,...,nmax+1 (imprimir)
do n=0,nmax*nfracs ! ensamblar matriz 4.26 y terminos independientes
M = z0; B = z0; iPIV = 6
```

$$\sigma_{rr1} = 2\mu_1 r^{-2} \sum_{n=0}^{\infty} \left[\varphi_0 \epsilon_n i^n \mathcal{E}_{11}^{(1)}(\alpha_1 r) + A_n \mathcal{E}_{11}^{(3)}(\alpha_1 r) + B_n \mathcal{E}_{12}^{(3)}(\beta_1 r) \right] \cos n\theta;$$

$$\sigma_{rr2} = 2\mu_2 r^{-2} \sum_{\underset{0}{n=2}}^{\infty} \left[C_n \mathcal{E}_{11}^{(3)}(\alpha_2 r) + D_n \mathcal{E}_{11}^{(4)}(\alpha_2 r) + M_n \mathcal{E}_{12}^{(3)}(\beta_2 r) + N_n \mathcal{E}_{12}^{(4)}(\beta_2 r) \right] \cos n\theta;$$

```
! sigma_{rr}1 = sigma_{rr}2   @ r = b
M(1,1) = - muR * e11(3,1,1,2,n)
M(1,2) = - muR * e12(3,2,1,2,n)
M(1,3) = e11(3,1,2,2,n)
M(1,4) = e11(4,1,2,2,n)
M(1,5) = e12(3,2,2,2,n)
M(1,6) = e12(4,2,2,2,n)
B(1,1) = (1.0) * et(n) * UI**n * muR * e11(1,1,1,2,n)
```

$$\sigma_{r\theta 1} = 2\mu_1 r^{-2} \sum_{n=0}^{\infty} \left[\varphi_0 \epsilon_n i^n \mathcal{E}_{41}^{(1)}(\alpha_1 r) + A_n \mathcal{E}_{41}^{(3)}(\alpha_1 r) + B_n \mathcal{E}_{42}^{(3)}(\beta_1 r) \right] \sin n\theta.$$

$$\sigma_{r\theta 2} = 2\mu_2 r^{-2} \sum_{n=0}^{\infty} \left[C_n \mathcal{E}_{41}^{(3)}(\alpha_2 r) + D_n \mathcal{E}_{41}^{(4)}(\alpha_2 r) + M_n \mathcal{E}_{42}^{(3)}(\beta_2 r) + N_n \mathcal{E}_{42}^{(4)}(\beta_2 r) \right] \sin n\theta.$$

```
! sigma_{r th}1 = sigma_{r th}2   @ r = b
M(2,1) = - muR * e41(3,1,1,2,n)
M(2,2) = - muR * e42(3,2,1,2,n)
M(2,3) = e41(3,1,2,2,n)
M(2,4) = e41(4,1,2,2,n)
M(2,5) = e42(3,2,2,2,n)
M(2,6) = e42(4,2,2,2,n)
B(2,1) = (1.0) * et(n) * UI**n * muR * e41(1,1,1,2,n)
```

$$u_{r1} = r^{-1} \sum_{n=0}^{\infty} \left[\varphi_o \epsilon_n i^n \mathcal{E}^{(1)}_{71}(\alpha_1 r) + A_n \mathcal{E}^{(3)}_{71}(\alpha_1 r) + B_n \mathcal{E}^{(3)}_{72}(\beta_1 r) \right] \cos n\theta;$$

(4.2

$$u_{r2} = r^{-1} \sum_{n=0}^{\infty} \left[C_n \mathcal{E}^{(3)}_{71}(\alpha_2 r) + D_n \mathcal{E}^{(4)}_{\underset{71}{81}}(\alpha_2 r) + M_n \mathcal{E}^{(3)}_{72}(\beta_2 r) \right.$$

$$\left. + N_n \mathcal{E}^{(4)}_{72}(\beta_2 r) \right] \cos n\theta;$$

```
! u_{r}1 = u_{r}2   @ r = b
M(3,1) = - e71(3,1,1,2,n)
M(3,2) = - e72(3,2,1,2,n)
M(3,3) = e71(3,1,2,2,n)
M(3,4) = e71(4,1,2,2,n)
M(3,5) = e72(3,2,2,2,n)
M(3,6) = e72(4,2,2,2,n)
B(3,1) = (1.0) * et(n) * UI**n * e71(1,1,1,2,n)
```

$$u_{\theta 1} = r^{-1} \sum_{n=0}^{\infty} \left[\varphi_o \epsilon_n i^n \mathcal{E}^{(1)}_{81}(\alpha_1 r) + A_n \mathcal{E}^{(3)}_{81}(\alpha_1 r) + B_n \mathcal{E}^{(3)}_{82}(\beta_1 r) \right] \sin n\theta.$$

(4.2

$$u_{\theta \underset{2}{1}} = r^{-1} \sum_{n=0}^{\infty} \left[C_n \mathcal{E}^{(3)}_{81}(\alpha_2 r) + D_n \mathcal{E}^{(4)}_{81}(\alpha_2 r) + M_n \mathcal{E}^{(3)}_{82}(\beta_2 r) \right.$$

$$\left. + N_n \mathcal{E}^{(4)}_{82}(\beta_2 r) \right] \sin n\theta,$$

```
! u_{th}1 = u_{th}2   @ r = b
M(4,1) = - e81(3,1,1,2,n)
M(4,2) = - e82(3,2,1,2,n)
M(4,3) = e81(3,1,2,2,n)
M(4,4) = e81(4,1,2,2,n)
```

```
M(4,5) = e82(3,2,2,2,n)
M(4,6) = e82(4,2,2,2,n)
B(4,1) = (1.0) * et(n) * UI**n * e81(1,1,1,2,n)
```

$$\sigma_{rr2} = 2\mu_2 r^{-2} \sum_{\substack{n=2\\0}}^{\infty} \Big[C_n \mathcal{E}_{11}^{(3)}(\alpha_2 r) + D_n \mathcal{E}_{11}^{(4)}(\alpha_2 r) + M_n \mathcal{E}_{12}^{(3)}(\beta_2 r) + N_n \mathcal{E}_{12}^{(4)}(\beta_2 r) \Big] \cos n\theta;$$

```
! sigma_{rr}2 = 0   @ r = a
M(5,1) = z0
M(5,2) = z0
M(5,3) = e11(3,1,2,1,n)
M(5,4) = e11(4,1,2,1,n)
M(5,5) = e12(3,2,2,1,n)
M(5,6) = e12(4,2,2,1,n)
B(5,1) = z0
```

$$\sigma_{r\theta 2} = 2\mu_2 r^{-2} \sum_{n=0}^{\infty} \Big[C_n \mathcal{E}_{41}^{(3)}(\alpha_2 r) + D_n \mathcal{E}_{41}^{(4)}(\alpha_2 r) + M_n \mathcal{E}_{42}^{(3)}(\beta_2 r) + N_n \mathcal{E}_{42}^{(4)}(\beta_2 r) \Big] \sin n\theta.$$

```
! sigma_{r th}2 = 0   @ r = a
M(6,1) = z0
M(6,2) = z0
M(6,3) = e41(3,1,2,1,n)
M(6,4) = e41(4,1,2,1,n)
M(6,5) = e42(3,2,2,1,n)
M(6,6) = e42(4,2,2,1,n)
B(6,1) = z0
```

by the following matrix

$$\begin{bmatrix} -\tilde{\mu}E_{11}^{(3)}(\alpha_1 b) & -\tilde{\mu}E_{12}^{(3)}(\beta_1 b) & E_{11}^{(3)}(\alpha_2 b) & E_{11}^{(4)}(\alpha_2 b) & E_{12}^{(3)}(\beta_2 b) & E_{12}^{(4)}(\beta_2 b) \\ -\tilde{\mu}E_{41}^{(3)}(\alpha_1 b) & -\tilde{\mu}E_{42}^{(3)}(\beta_1 b) & E_{41}^{(3)}(\alpha_2 b) & E_{41}^{(4)}(\alpha_2 b) & E_{42}^{(3)}(\beta_2 b) & E_{42}^{(4)}(\beta_2 b) \\ (-)E_{71}^{(3)}(\alpha_1 b) & (-)E_{72}^{(3)}(\beta_1 b) & E_{71}^{(3)}(\alpha_2 b) & E_{71}^{(4)}(\alpha_2 b) & E_{72}^{(3)}(\beta_2 b) & E_{72}^{(4)}(\beta_2 b) \\ (-)E_{81}^{(3)}(\alpha_1 b) & (-)E_{82}^{(3)}(\beta_1 b) & E_{81}^{(3)}(\alpha_2 b) & E_{81}^{(4)}(\alpha_2 b) & E_{82}^{(3)}(\beta_2 b) & E_{82}^{(4)}(\beta_2 b) \\ 0 & 0 & E_{11}^{(3)}(\alpha_2 \overset{a}{b}) & E_{11}^{(4)}(\alpha_2 \overset{a}{b}) & E_{12}^{(3)}(\beta_2 a) & E_{12}^{(4)}(\beta_2 a) \\ 0 & 0 & \underset{41}{E_{21}^{(3)}(\alpha_2 a)} & \underset{41}{E_{21}^{(4)}(\alpha_2 a)} & \underset{42}{E_{22}^{(3)}(\beta_2 a)} & \underset{42}{E_{22}^{(4)}(\beta_2 a)} \end{bmatrix} \begin{bmatrix} A_n \\ B_n \\ C_n \\ D_n \\ M_n \\ N_n \end{bmatrix} = \overset{(+)}{-}\varphi_o \epsilon_n i^n \begin{bmatrix} \tilde{\mu}E_{11}^{(1)}(\alpha_1 b) \\ \tilde{\mu}E_{41}^{(1)}(\alpha_1 b) \\ E_{71}^{(1)}(\alpha_1 b) \\ E_{81}^{(1)}(\alpha_1 b) \\ 0 \end{bmatrix}$$

```
      call zgesv(6,1,M(1:6,1:6),6,IPIV,B(:,1),6,info)
      if(info .ne. 0) then
        write(6,'(A,I0,a,I0)', ADVANCE = "NO") &
        "se corta la suma en ",n, "system info = ",info
        exit
      else if (abs(B(1,1)) .lt. 0.00000001) then  !NaN
       write(6,'(A,I0,a)', ADVANCE = "NO") &
      "se corta la suma en ",n, " por chiquito"
       exit
      end if
!     call showMNmatrixZ(6,1,B,"  A  ",6)
```

```
      !elementos mecanicos
      do i = 1,nRes
       r => Rw(i)%r
       th => Rw(i)%th
       reg => Rw(i)%reg
        if (reg .eq. 1) then
          Rw(i)%s_rr_1(J) = Rw(i)%s_rr_1(J) + &
          (1. * et(n) * UI**n * e11(1,1,1,2+i,n) + &
                       B(1,1) * e11(3,1,1,2+i,n) + &
                       B(2,1) * e12(3,2,1,2+i,n)) * (cos(n * th))
          Rw(i)%s_tt_1(J) = Rw(i)%s_tt_1(J) + &
          (1. * et(n) * UI**n * e21(1,1,1,2+i,n) + &
                       B(1,1) * e21(3,1,1,2+i,n) + &
                       B(2,1) * e22(3,2,1,2+i,n)) * (cos(n * th))
          Rw(i)%s_rt_1(J) = Rw(i)%s_rt_1(J) + &
          (1. * et(n) * UI**n * e41(1,1,1,2+i,n) + &
                       B(1,1) * e41(3,1,1,2+i,n) + &
                       B(2,1) * e42(3,2,1,2+i,n)) * (sin(n * th))
          Rw(i)%u_r_1(J) = Rw(i)%u_r_1(J) + &
          (1. * et(n) * UI**n * e71(1,1,1,2+i,n) + &
```

```
                 B(1,1) * e71(3,1,1,2+i,n) + &
                 B(2,1) * e72(3,2,1,2+i,n)) * (cos(n * th))
    Rw(i)%u_t_1(J) = Rw(i)%u_t_1(J) + &
    (1. * et(n) * UI**n * e81(1,1,1,2+i,n) + &
                 B(1,1) * e81(3,1,1,2+i,n) + &
                 B(2,1) * e82(3,2,1,2+i,n)) * (sin(n * th))
   else if (reg .eq. 2) then
    Rw(i)%s_rr_2(J) = Rw(i)%s_rr_2(J) + &
                (B(3,1) * e11(3,1,2,2+i,n) + &
                 B(4,1) * e11(4,1,2,2+i,n) + &
                 B(5,1) * e12(3,2,2,2+i,n) + &
                 B(6,1) * e12(4,2,2,2+i,n)) * (cos(n * th))
    Rw(i)%s_tt_2(J) = Rw(i)%s_tt_2(J) + &
                (B(3,1) * e21(3,1,2,2+i,n) + &
                 B(4,1) * e21(4,1,2,2+i,n) + &
                 B(5,1) * e22(3,2,2,2+i,n) + &
                 B(6,1) * e22(4,2,2,2+i,n)) * (cos(n * th))
    Rw(i)%s_rt_2(J) = Rw(i)%s_rt_2(J) + &
                (B(3,1) * e41(3,1,2,2+i,n) + &
                 B(4,1) * e41(4,1,2,2+i,n) + &
                 B(5,1) * e42(3,2,2,2+i,n) + &
                 B(6,1) * e42(4,2,2,2+i,n)) * (sin(n * th))
    Rw(i)%u_r_2(J) = Rw(i)%u_r_2(J) + &
                (B(3,1) * e71(3,1,2,2+i,n) + &
                 B(4,1) * e71(4,1,2,2+i,n) + &
                 B(5,1) * e72(3,2,2,2+i,n) + &
                 B(6,1) * e72(4,2,2,2+i,n)) * (cos(n * th))
    Rw(i)%u_t_2(J) = Rw(i)%u_t_2(J) + &
                (B(3,1) * e81(3,1,2,2+i,n) + &
                 B(4,1) * e81(4,1,2,2+i,n) + &
                 B(5,1) * e82(3,2,2,2+i,n) + &
                 B(6,1) * e82(4,2,2,2+i,n)) * (sin(n * th))
   end if
  end do! i:nRes
  end do !n

  ! términos fuera de la suma:
  do i = 1,nRes
   r => Rw(i)%r
   th => Rw(i)%th
   reg => Rw(i)%reg
   sig0 = amu(1) * beta(1)**2. !eq 3.15
    if (reg .eq. 1) then
     Rw(i)%s_rr_1(J) = Rw(i)%s_rr_1(J) * 2. *amu(1) / r**2.
     Rw(i)%s_tt_1(J) = Rw(i)%s_tt_1(J) * 2. *amu(1) / r**2. / sig0
     Rw(i)%s_rt_1(J) = Rw(i)%s_rt_1(J) * 2. *amu(1) / r**2.
     Rw(i)%u_r_1(J) = Rw(i)%u_r_1(J) / r
     Rw(i)%u_t_1(J) = Rw(i)%u_t_1(J) / r
    else if (reg .eq. 2) then
     Rw(i)%s_rr_2(J) = Rw(i)%s_rr_2(J) * 2. *amu(2) / r**2.
     Rw(i)%s_tt_2(J) = Rw(i)%s_tt_2(J) * 2. *amu(2) / r**2. / sig0
     Rw(i)%s_rt_2(J) = Rw(i)%s_rt_2(J) * 2. *amu(2) / r**2.
```

```
    Rw(i)%u_r_2(J) = Rw(i)%u_r_2(J) / r
    Rw(i)%u_t_2(J) = Rw(i)%u_t_2(J) / r
  end if
end do! i:nRes
```

```
abscisa(J) = dfrec * 2 * pi * J * radios(1) / cp(1)
end do !J
! plot curves
! end program tunelRecub
```

Se comparan los resultados normalizados para $\eta = 1{,}1$, $r = b$,



Fig. 4.5a. Magnitudes of the Tangential Stress of the Medium at the Boundary of the Liner at θ = π/2, π

en $\theta = \pi/2$ y en $\theta = \pi$ respectivamente,





y en $\eta = 1{,}1$, $r = a$,



Fig. 4.5b. Magnitudes of the Tangential Stress at the Inner Side of the Cylinder at θ = π/2, π

en $\theta = \pi/2$ y en $\theta = \pi$ respectivamente,





```
function e11(i,c,reg,r,n)
use vars_func_of_w, only : alfa,s2,gammaP
use datos, only : radios
use Hank, only : Bess
implicit none
complex*16 :: e11
integer :: i,c,reg,r,n
complex*16, pointer :: Bess_n,Bess_n_1
```

```
Bess_n => Bess(reg)%JYH1H2(i,n)%r(r)%c(c)
Bess_n_1 => Bess(reg)%JYH1H2(i,n-1)%r(r)%c(c)
e11 = (n**2 + n - s2(reg) * radios(r)**2 / 2. + &
gammaP**2 * radios(r)**2) * Bess_n - alfa(reg)* radios(r) * Bess_n_1
end function e11
```

ψ:
$$\frac{2\mu}{r^2}\left[r\,\frac{\partial^2\psi}{\partial r\partial\theta} - \frac{\partial\psi}{\partial\theta}\right] = \left(\frac{2\mu}{r^2}\right)e_{12}^{(i)}\begin{Bmatrix}\sin n\theta\\ \boxed{\cos n\theta}\end{Bmatrix} e^{\pm i\gamma_s z},$$

$$e_{12}^{(i)} = \mp n\left[-\ (n+1)\ \mathfrak{C}_n^{(i)}(\beta r) + \beta r\ \mathfrak{C}_{n-1}^{(i)}(\beta r)\right].$$

```
function e12(i,c,reg,r,n)
use vars_func_of_w, only : beta
use datos, only : radios
use Hank, only : Bess
implicit none
complex*16 :: e12
integer :: i,c,reg,r,n
complex*16, pointer :: Bess_n,Bess_n_1
Bess_n => Bess(reg)%JYH1H2(i,n)%r(r)%c(c)
Bess_n_1 => Bess(reg)%JYH1H2(i,n-1)%r(r)%c(c)
e12 = n * (-(n+1)* Bess_n + beta(reg) * radios(r) * Bess_n_1)
end function e12
```

φ:
$$\lambda\nabla^2\varphi + 2\mu\left(\frac{1}{r}\frac{\partial\varphi}{\partial r} + \frac{1}{r^2}\frac{\partial^2\varphi}{\partial\theta^2}\right) = \frac{2\mu}{r}\,e_{21}^{(i)}\begin{Bmatrix}\boxed{\cos n\theta}\\ \sin n\theta\end{Bmatrix} e^{\pm i\gamma_p z},$$

$$e_{21}^{(i)} = -\ (n^2 + n + s^2r^2/2 - p^2r^2)\ \mathfrak{C}_n^{(i)}(\alpha r) + \alpha r\ \mathfrak{C}_{n-1}^{(i)}(\alpha r).$$

```
function e21(i,c,reg,r,n)
use vars_func_of_w, only : alfa,s2,p2
use datos, only : radios
use Hank, only : Bess
implicit none
complex*16 :: e21
integer :: i,c,reg,r,n
complex*16, pointer :: Bess_n,Bess_n_1
Bess_n => Bess(reg)%JYH1H2(i,n)%r(r)%c(c)
Bess_n_1 => Bess(reg)%JYH1H2(i,n-1)%r(r)%c(c)
e21 = - (n**2 + n + s2(reg)*radios(r)**2 / 2. - p2(reg)*radios(r)**2) * &
       Bess_n + alfa(reg) * radios(r) * Bess_n_1
```

```
end function e21
```

$$\psi: \qquad \frac{2\mu}{r^2}\left[\frac{\partial\psi}{\partial\theta} - r\,\frac{\partial^2\psi}{\partial r\,\partial\theta}\right] = \frac{2\mu}{r^2}\; e_{22}^{(i)} \begin{Bmatrix} \sin n\theta \\ \boxed{\cos n\theta} \end{Bmatrix} e^{\pm i\gamma_s z},$$

$$e_{22}^{(i)} = \mp n\left[(n+1)\,\mathfrak{C}_n^{(i)}(\beta r) - \beta r\,\mathfrak{C}_{n-1}^{(i)}(\beta r)\right].$$

```
function e22(i,c,reg,r,n)
use vars_func_of_w, only : beta
use datos, only : radios
use Hank, only : Bess
implicit none
complex*16 :: e22
integer :: i,c,reg,r,n
complex*16, pointer :: Bess_n,Bess_n_1
Bess_n => Bess(reg)%JYH1H2(i,n)%r(r)%c(c)
Bess_n_1 => Bess(reg)%JYH1H2(i,n-1)%r(r)%c(c)
e22 = n * ((n+1)*Bess_n - beta(reg)*radios(r)*Bess_n_1)
end function e22
```

$$\varphi: \qquad \frac{2\mu}{r^2}\left[r\,\frac{\partial^2\varphi}{\partial r\,\partial\theta} - \frac{\partial\varphi}{\partial\theta}\right] = \frac{2\mu}{r^2}\; e_{41}^{(i)} \begin{Bmatrix} \boxed{\sin n\theta} \\ \cos n\theta \end{Bmatrix} e^{\pm i\gamma_p z},$$

$$e_{41}^{(i)} = \mp n\left[-\,(n+1)\,\mathfrak{C}_n^{(i)}(\alpha r) + \alpha r\,\mathfrak{C}_{n-1}^{(i)}(\alpha r)\right].$$

```
function e41(i,c,reg,r,n)
use vars_func_of_w, only : alfa
use datos, only : radios
use Hank, only : Bess
implicit none
complex*16 :: e41
integer :: i,c,reg,r,n
complex*16, pointer :: Bess_n,Bess_n_1
Bess_n => Bess(reg)%JYH1H2(i,n)%r(r)%c(c)
Bess_n_1 => Bess(reg)%JYH1H2(i,n-1)%r(r)%c(c)
e41 = - n * (-(n+1) * Bess_n + alfa(reg)*radios(r)*Bess_n_1)
end function e41
```

ψ:
$$\frac{\mu}{r^2}\left[\frac{\partial^2\psi}{\partial\theta^2} - r^3\frac{\partial}{\partial r}\left(\frac{1}{r}\frac{\partial\psi}{\partial r}\right)\right] = \frac{2\mu}{r^2}\,e_{42}^{(i)}\begin{Bmatrix}\cos n\theta\\ \boxed{\sin n\theta}\end{Bmatrix}e^{\pm i\gamma_s z},$$

$$e_{42}^{(i)} = -\left(n^2 + n - \frac{\beta^2 r^2}{2}\right)\mathfrak{C}_n^{(i)}(\beta r) + \beta r\,\mathfrak{C}_{n-1}^{(i)}(\beta r).$$

```
function e42(i,c,reg,r,n)
use vars_func_of_w, only : beta
use datos, only : radios
use Hank, only : Bess
implicit none
complex*16 :: e42
integer :: i,c,reg,r,n
complex*16, pointer :: Bess_n,Bess_n_1
Bess_n => Bess(reg)%JYH1H2(i,n)%r(r)%c(c)
Bess_n_1 => Bess(reg)%JYH1H2(i,n-1)%r(r)%c(c)
e42 = - (n**2 + n - beta(reg)**2 * radios(r)**2 / 2.) * Bess_n + &
      beta(reg) * radios(r) * Bess_n_1
end function e42
```

φ:
$$\frac{\partial\varphi}{\partial r} = \frac{1}{r}\,e_{71}^{(i)}\begin{Bmatrix}\boxed{\cos n\theta}\\ \sin n\theta\end{Bmatrix}e^{\pm i\gamma_p z},$$

$$e_{71}^{(i)} = \left[\alpha r\,\mathfrak{C}_{n-1}^{(i)}(\alpha r) - n\,\mathfrak{C}_n^{(i)}(\alpha r)\right].$$

```
function e71(i,c,reg,r,n)
use vars_func_of_w, only : alfa
use datos, only : radios
use Hank, only : Bess
implicit none
complex*16 :: e71
integer :: i,c,reg,r,n
complex*16, pointer :: Bess_n,Bess_n_1
Bess_n => Bess(reg)%JYH1H2(i,n)%r(r)%c(c)
Bess_n_1 => Bess(reg)%JYH1H2(i,n-1)%r(r)%c(c)
e71 = alfa(reg) * radios(r) * Bess_n_1 - n * Bess_n
end function e71
```

ψ:
$$\frac{1}{r}\frac{\partial\psi}{\partial\theta} = \frac{1}{r}\,e_{72}^{(i)}\begin{Bmatrix}\sin n\theta\\ \boxed{\cos n\theta}\end{Bmatrix}e^{\pm i\gamma_s z},$$

$$e_{72}^{(i)} = \mp n\,\mathfrak{C}_n^{(i)}(\beta r).$$

```
function e72(i,c,reg,r,n)
use Hank, only : Bess
implicit none
complex*16 :: e72
integer :: i,c,reg,r,n
complex*16, pointer :: Bess_n
Bess_n => Bess(reg)%JYH1H2(i,n)%r(r)%c(c)
e72 = n * Bess_n
end function e72
```

φ: $$\frac{1}{r}\frac{\partial\varphi}{\partial\theta} = \frac{1}{r}\; e_{81}^{(i)} \left\{ \begin{array}{c} \boxed{\sin n\theta} \\ \cos n\theta \end{array} \right\} e^{\pm i\gamma_p z},$$

$$e_{81}^{(i)} = \mp n\, \mathfrak{C}_n^{(i)}(\alpha r).$$

```
function e81(i,c,reg,r,n)
use Hank, only : Bess
implicit none
complex*16 :: e81
integer :: i,c,reg,r,n
complex*16, pointer :: Bess_n
Bess_n => Bess(reg)%JYH1H2(i,n)%r(r)%c(c)
e81 = - n * Bess_n
end function e81
```

ψ: $$-\frac{\partial\psi}{\partial r} = \frac{1}{r}\; e_{82}^{(i)} \left\{ \begin{array}{c} \cos n\theta \\ \boxed{\sin n\theta} \end{array} \right\} e^{\pm i\gamma_s z},$$

$$e_{82}^{(i)} = -\left[\beta r\, \mathfrak{C}_{n-1}^{(i)}(\beta r) - n\, \mathfrak{C}_n^{(i)}(\beta r)\right].$$

```
function e82(i,c,reg,r,n)
use vars_func_of_w, only : beta
use datos, only : radios
use Hank, only : Bess
implicit none
complex*16 :: e82
integer :: i,c,reg,r,n
complex*16, pointer :: Bess_n,Bess_n_1
Bess_n => Bess(reg)%JYH1H2(i,n)%r(r)%c(c)
Bess_n_1 => Bess(reg)%JYH1H2(i,n-1)%r(r)%c(c)
```

```
e82 = - (beta(reg) * radios(r) * Bess_n_1 &
        - n * Bess_n)
end function e82
```